



# # Competitive Security Assessment

ParaSpace cAPE P2

Mar 26th, 2023

---

Summary	3
Overview	4
Audit Scope	5
Code Assessment Findings	6
PSC-1:Additional yield does not match bufferStakingBalance	8
PSC-2:Did not judge the actual payment reward	9
PSC-3:Potential Reentrancy Attack	11
PSC-4:Unchecked Return Value	12
PSC-5:Unused return value	14
PSC-6:Use of unified fund withdrawal function	17
PSC-7: <code>_getTotalPooledApeBalance</code> may be manipulated by <code>rewardAmount</code>	18
PSC-8: <code>stakingBalance</code> should use <code>realWithdraw</code> value, not use the input param <code>amount</code>	20
PSC-9: <code>tmp_fix_withdrawFromApeCoinStaking</code> should only be allowed once	22
PSC-10:event duplicate in <code>AutoCompoundApe.sol</code>	24
PSC-11:liquidate the hacker for profit	30
PSC-12: <code>tmp_fix_withdrawFromApeCoinStaking</code> may be suffering from front-run attack	31
Disclaimer	33

## Summary

This report is prepared for the project to identify vulnerabilities and issues in the smart contract source code. A group of NDA covered experienced security experts have participated in the Secure3's Audit Contest to find vulnerabilities and optimizations. Secure3 team has participated in the contest process as well to provide extra auditing coverage and scrutiny of the finding submissions.

The comprehensive examination and auditing scope includes:

- Cross checking contract implementation against functionalities described in the documents and white paper disclosed by the project owner.
- Contract Privilege Role Review to provide more clarity on smart contract roles and privilege.
- Using static analysis tools to analyze smart contracts against common known vulnerabilities patterns.
- Verify the code base is compliant with the most up-to-date industry standards and security best practices.
- Comprehensive line-by-line manual code review of the entire codebase by industry experts.

The security assessment resulted in findings that are categorized in four severity levels: Critical, Medium, Low, Informational. For each of the findings, the report has included recommendations of fix or mitigation for security and best practices.

# Overview

## Project Detail

<b>Project Name</b>	ParaSpace cAPE P2
<b>Platform &amp; Language</b>	Solidity
<b>Codebase</b>	<ul style="list-style-type: none"><li>• <a href="https://github.com/para-space/paraspace-core">https://github.com/para-space/paraspace-core</a></li><li>• audit commit - 684dd70b9c76c47c9742ae4bbcf4d645090c58cf</li><li>• final commit - d3263e22565e7715c12a145313a615cde50a03fc</li></ul>
<b>Audit Methodology</b>	<ul style="list-style-type: none"><li>• Audit Contest</li><li>• Business Logic and Code Review</li><li>• Privileged Roles Review</li><li>• Static Analysis</li></ul>


## Code Vulnerability Review Summary

Vulnerability Level	Total	Reported	Acknowledged	Fixed	Mitigated	Declined
<b>Critical</b>	0	0	0	0	0	0
<b>Medium</b>	4	0	1	2	0	1
<b>Low</b>	5	1	3	0	0	1
<b>Informational</b>	3	0	2	0	0	1

## Audit Scope

File	Commit Hash
contracts/misc/AutoCompoundApe.sol	684dd70b9c76c47c9742ae4bbcf4d645090c58cf
contracts/protocol/libraries/logic/ReserveLogic.sol	684dd70b9c76c47c9742ae4bbcf4d645090c58cf
contracts/protocol/libraries/logic/SupplyLogic.sol	684dd70b9c76c47c9742ae4bbcf4d645090c58cf
contracts/protocol/libraries/logic/ValidationLogic.sol	684dd70b9c76c47c9742ae4bbcf4d645090c58cf
contracts/protocol/pool/PoolCore.sol	684dd70b9c76c47c9742ae4bbcf4d645090c58cf

## Code Assessment Findings



ID	Name	Category	Severity	Status	Contributor
PSC-1	Additional yield does not match bufferStakingBalance	Logic	Medium	Fixed	Kong7ych3
PSC-2	Did not judge the actual payment reward	Logical	Medium	Fixed	Hupixiong3
PSC-3	Potential Reentrancy Attack	Reentrancy	Low	Acknowledged	BradMoonU ESTC
PSC-4	Unchecked Return Value	Logical	Medium	Declined	BradMoonU ESTC

PSC-5	Unused return value	Code Style	Informational	Acknowledged	BradMoonU ESTC
PSC-6	Use of unified fund withdrawal function	Logical	Low	Declined	Hupixiong3
PSC-7	<code>_getTotalPooledApeBalance</code> may be manipulated by <code>rewardAmount</code>	Logical	Medium	Acknowledged	Secure3
PSC-8	<code>stakingBalance</code> should use <code>realWithdraw</code> value, not use the input param <code>amount</code>	Code Style	Informational	Declined	xfu
PSC-9	<code>tmp_fix_withdrawFromApeCoinStaking</code> should only be allowed once	Logical	Low	Acknowledged	Secure3
PSC-10	event duplicate in <code>AutoCompoundApe.sol</code>	Gas Optimization	Informational	Acknowledged	xfu
PSC-11	liquidate the hacker for profit	Logical	Low	Acknowledged	comcat
PSC-12	<code>tmp_fix_withdrawFromApeCoinStaking</code> may be suffering from front-run attack	Logical	Low	Reported	thereksfour

## PSC-1:Additional yield does not match bufferStakingBalance

Category	Severity	Code Reference	Status	Contributor
Logic	Medium	<ul style="list-style-type: none"><li>code/contracts/misc/AutoCompoundApe.sol#L111</li></ul>	Fixed	Kong7ych3

### Code

```
111:         stakingBalance -= amount;
```

### Description

**Kong7ych3** : In the temporary repair function for exchange rate, a fixed amount is used for withdrawal to update `bufferStakingBalance`. If other users still stake for the cAPE contract before the repair is completed, the cAPE contract will receive additional benefits. This part of the income will be distributed to users, so the amount that all users can withdraw in cAPE will be larger than expected. This should be a good thing, but unfortunately `bufferStakingBalance` will not be updated accordingly, which will cause a user to revert in the future when performing the `_withdrawFromApeCoinStaking` operation because the withdrawn amount will be greater than `bufferStakingBalance`, resulting in user funds being locked.

```
function _withdrawFromApeCoinStaking(uint256 amount) internal {  
    ...  
    bufferStakingBalance -= amount;  
    ...  
}
```

### Recommendation

**Kong7ych3** : It is recommended to add a function to update the `bufferStakingBalance` parameter to avoid the above issue.

Consider below fix:

```
function rebaseFromApeCoinStaking() external onlyOwner {  
    (bufferStakingBalance, ) = apeStaking.addressPosition(address(this));  
}
```

### Client Response

Fixed.



## PSC-2: Did not judge the actual payment reward

Category	Severity	Code Reference	Status	Contributor
Logical	Medium	- code/contracts/misc/AutoCompound Ape.sol#L134-L144	Fixed	Hupixiong3

### Code

```
134:     function _harvest() internal {
135:         uint256 rewardAmount = apeStaking.pendingRewards(
136:             APE_COIN_POOL_ID,
137:             address(this),
138:             0
139:         );
140:         if (rewardAmount > 0) {
141:             apeStaking.claimSelfApeCoin();
142:             bufferBalance += rewardAmount;
143:         }
144:     }
```

### Description

**Hupixiong3** : When the reward is claimed through the `_harvest()` function, the contract balance change is not determined to be consistent with `rewardAmount`. When there is any change or malfunction to the pledge agreement of APE, `bufferBalance` will update error.

### Recommendation

**Hupixiong3** : Adding a judgment about whether the contract balance changes in line with `rewardAmount` when the reward is claimed through the `_harvest()` function can effectively prevent errors.

Consider below fix in the `AutoCompoundApe._harvest()` function

```
function _harvest() internal {
    uint256 rewardAmount = apeStaking.pendingRewards(
        APE_COIN_POOL_ID,
        address(this),
        0
    );
    if (rewardAmount > 0) {
        uint256 balanceBefore = apeCoin.balanceOf(address(this));
        apeStaking.claimSelfApeCoin();
        uint256 balanceAfter = apeCoin.balanceOf(address(this));
        uint256 realClaim = balanceAfter - balanceBefore;
        require(rewardAmount==realClaim,"Reward error")
        bufferBalance += rewardAmount;
    }
}
```

## Client Response

Fixed.

## PSC-3:Potential Reentrancy Attack

Category	Severity	Code Reference	Status	Contributor
Reentrancy	Low	<ul style="list-style-type: none"><li>code/contracts/misc/AutoCompoundApe.sol#L14</li></ul>	Acknowledged	BradMoonUESTC

### Code

```
14:contract AutoCompoundApe is
```

### Description

**BradMoonUESTC** : In the contract autocompoundape, there are a large number of potential token transfers. In this case, if there is a received hook logic, the attacker may use this to carry out re-entry attacks

### Recommendation

**BradMoonUESTC** : Use reentrancy lock

### Client Response

Acknowledged.We did not fix it because we think ApeCoin don't have received hook logic.

## PSC-4:Unchecked Return Value

Category	Severity	Code Reference	Status	Contributor
Logical	Medium	<ul style="list-style-type: none"><li>code/contracts/misc/AutoCompoundApe.sol#L125-L145</li></ul>	Declined	BradMoonUESTC

### Code

```
125:     function _compound() internal {
126:         uint256 _bufferBalance = bufferBalance;
127:         if (_bufferBalance >= MIN_OPERATION_AMOUNT) {
128:             apeStaking.depositSelfApeCoin(_bufferBalance);
129:             stakingBalance += _bufferBalance;
130:             bufferBalance = 0;
131:         }
132:     }
133:
134:     function _harvest() internal {
135:         uint256 rewardAmount = apeStaking.pendingRewards(
136:             APE_COIN_POOL_ID,
137:             address(this),
138:             0
139:         );
140:         if (rewardAmount > 0) {
141:             apeStaking.claimSelfApeCoin();
142:             bufferBalance += rewardAmount;
143:         }
144:     }
145:
```

### Description

**BradMoonUESTC** : The AutoCompoundApe contract implements an automatic compounding mechanism for the Ape token by using the ApeCoinStaking contract. The contract has functions for depositing, withdrawing, harvesting rewards, and compounding the Ape token. However, there are potential security issues with the compound and harvest functions, which may result in incorrect updates to the bufferBalance variable.

The compound function uses the depositSelfApeCoin function of the ApeCoinStaking contract to deposit Ape tokens and updates the bufferBalance variable. However, if the deposit operation fails or returns an unexpected value, the

bufferBalance variable may not be updated correctly. Similarly, the harvest function uses the claimSelfApeCoin function to claim rewards and updates the bufferBalance variable, which may also be affected by unexpected returns or failures.

## Recommendation

**BradMoonUESTC** : To address these potential security issues, we recommend adding error handling and checks in the compound and harvest functions to ensure that the deposit and claim operations are successful and return the expected values. Additionally, it is essential to monitor the ApeCoinStaking contract for any changes or potential vulnerabilities that may affect the functionality of the AutoCompoundApe contract. Furthermore, it is recommended to use the ReentrancyGuard to prevent re-entrancy attacks and to follow best practices for secure coding and contract development. Finally, it is essential to test the contract thoroughly and perform audits by security experts to ensure its safety and reliability.

## Client Response

Declined. These functions don't have a return value.

## PSC-5:Unused return value

Category	Severity	Code Reference	Status	Contributor
Code Style	Informational	<ul style="list-style-type: none"><li>code/contracts/protocol/libraries/Logic/SupplyLogic.sol#L118-L122</li><li>code/contracts/protocol/pool/PoolCore.sol#L504-L522</li><li>code/contracts/protocol/pool/PoolCore.sol#L535-L553</li></ul>	Acknowledged	BradMoonUES TC

## Code

```
118:         IVariableDebtToken(debtTokenAddress).burn(
119:             from,
120:             debtBalance,
121:             borrowIndex
122:         );

504:     LiquidationLogic.executeLiquidateERC20(
505:         ps._reserves,
506:         ps._reservesList,
507:         ps._usersConfig,
508:         DataTypes.ExecuteLiquidateParams({
509:             reservesCount: ps._reservesCount,
510:             liquidationAmount: liquidationAmount,
511:             auctionRecoveryHealthFactor: ps._auctionRecoveryHealthFactor,
512:             weth: ADDRESSES_PROVIDER.getWETH(),
513:             collateralAsset: collateralAsset,
514:             liquidationAsset: liquidationAsset,
515:             borrower: borrower,
516:             liquidator: msg.sender,
517:             receiveXToken: receivePToken,
518:             priceOracle: ADDRESSES_PROVIDER.getPriceOracle(),
519:             priceOracleSentinel: ADDRESSES_PROVIDER.getPriceOracleSentinel(),
520:             collateralTokenId: 0
521:         })
522:     );

535:     LiquidationLogic.executeLiquidateERC721(
536:         ps._reserves,
537:         ps._reservesList,
538:         ps._usersConfig,
539:         DataTypes.ExecuteLiquidateParams({
540:             reservesCount: ps._reservesCount,
541:             liquidationAmount: maxLiquidationAmount,
542:             auctionRecoveryHealthFactor: ps._auctionRecoveryHealthFactor,
543:             weth: ADDRESSES_PROVIDER.getWETH(),
544:             collateralAsset: collateralAsset,
545:             liquidationAsset: ADDRESSES_PROVIDER.getWETH(),
546:             collateralTokenId: collateralTokenId,
547:             borrower: borrower,
548:             liquidator: msg.sender,
549:             receiveXToken: receiveNToken,
550:             priceOracle: ADDRESSES_PROVIDER.getPriceOracle(),
```

```
551:         priceOracleSentinel: ADDRESSES_PROVIDER.getPriceOracleSentinel()  
552:     })  
553: );
```

## Description

**BradMoonUESTC** : Not using the return values from these functions can potentially lead to issues with error handling or tracking the state of the tokens being burned or minted. For example, if the burn function call were to fail for some reason, such as insufficient balance or an invalid input, the function would throw an exception, but since the return value is not being used, the exception would not be caught or handled in any way.

It is generally considered good coding practice to handle the return values of functions appropriately, as it helps to ensure the correct functioning of the code and improve its robustness.

## Recommendation

**BradMoonUESTC** : Ensure the return value of external function calls is used. Remove or comment out the unused return function parameters.

## Client Response

Acknowledged. We think it's ok.



## PSC-6:Use of unified fund withdrawal function

Category	Severity	Code Reference	Status	Contributor
Logical	Low	<ul style="list-style-type: none"><li>code/contracts/misc/AutoCompoundApe.sol#L200</li></ul>	Declined	Hupixiong3

### Code

```
200:         apeStaking.withdrawApeCoin(amount, receiver);
```

### Description

**Hupixiong3** : No uniform fund withdrawal function is used,tmp\_fix\_withdrawFromApeCoinStaking() function and \_withdrawFromApeCoinStaking() function money back call interface not consistent.

### Recommendation

**Hupixiong3** : Use of unified fund withdrawal function

Consider below fix in the `AutoCompoundApe.tmp_fix_withdrawFromApeCoinStaking()` function

```
function tmp_fix_withdrawFromApeCoinStaking(address receiver)
    external
    onlyOwner
{
    uint256 amount = 2332214464588784613678467;
    apeStaking.withdrawSelfApeCoin(amount, receiver);
    (stakingBalance, ) = apeStaking.addressPosition(address(this));
}
```

### Client Response

The recommendation is not right. withdrawSelfApecoin don't have a receiver parameter.

## PSC-7: `_getTotalPooledApeBalance` may be manipulated by `rewardAmount`

Category	Severity	Code Reference	Status	Contributor
Logical	Medium	<ul style="list-style-type: none"><li><code>code/contracts/misc/AutoCompoundApe.sol#L92-L104</code></li></ul>	Acknowledged	Secure3

### Code

```
92:     function _getTotalPooledApeBalance()
93:         internal
94:         view
95:         override
96:         returns (uint256)
97:     {
98:         uint256 rewardAmount = apeStaking.pendingRewards(
99:             APE_COIN_POOL_ID,
100:            address(this),
101:            0
102:        );
103:         return stakingBalance + rewardAmount + bufferBalance;
104:     }
```

### Description

**Secure3**: In the `_getTotalPooledApeBalance()` function, the `rewardAmount` is calculated by `apeStaking.pendingRewards()` call.

The `APE_COIN_POOL_ID` is a constant 0, and the parameter `_address` is the address of `AutoCompoundApe` contract

```
uint256 rewardAmount = apeStaking.pendingRewards(
    APE_COIN_POOL_ID,
    address(this),
    0
);
}
```

Zoom into `apeStaking.pendingRewards()` function (in the dependency and a Yuga contract)

```
function pendingRewards(uint256 _poolId, address _address, uint256 _tokenId) external view returns
(uint256) {
    Pool memory pool = pools[_poolId];
    Position memory position = _poolId == 0 ? addressPosition[_address]: nftPosition[_poolId]
[_tokenId];

    (uint256 rewardsSinceLastCalculated,) = rewardsBy(_poolId, pool.lastRewardedTimestampHour,
getPreviousTimestampHour());
    uint256 accumulatedRewardsPerShare = pool.accumulatedRewardsPerShare;

    if (block.timestamp > pool.lastRewardedTimestampHour + SECONDS_PER_HOUR && pool.stakedAmount
!= 0) {
        accumulatedRewardsPerShare = accumulatedRewardsPerShare + rewardsSinceLastCalculated *
APE_COIN_PRECISION / pool.stakedAmount;
    }
    return ((position.stakedAmount * accumulatedRewardsPerShare).toInt256() -
position.rewardsDebt).toUint256() / APE_COIN_PRECISION;
}
```

Because `APE_COIN_POOL_ID` is 0, `position` is `addressPosition[_address]`. This is the same cause of the previous hack. `accumulatedRewardsPerShare` does not increase in one block because the `depositApeCoin()` operation calls `updatePool(APECOIN_POOL_ID)` and ensures `pool.lastRewardedTimestampHour` is updated. The `_recipient` parameter in `depositApeCoin()` is set to the `AutoCompoundApe`'s address, so `position.stakedAmount` may be manipulated. This may result in `_getTotalPooledApeBalance()` gets manipulated.

However, since the value of `accumulatedRewardsPerShare` is unclear, this maybe very difficult for hacker to borrow more than the amount staked into the contract within the same block.

## Recommendation

**Secure3** : To be safe, need a storage variable `rewardAmount` to record the reward amount in the `AutoCompoundApe`.

## Client Response

Acknowledged. We think it's impossible that `rewardAmount` can be manipulated.

## PSC-8: `stakingBalance` should use `realWithdraw` value, not use the input param `amount`

Category	Severity	Code Reference	Status	Contributor
Code Style	Informational	<ul style="list-style-type: none"><li><code>code/contracts/misc/AutoCompoundApe.sol#L106-L113</code></li></ul>	Declined	xfu

### Code

```
106: function _withdrawFromApeCoinStaking(uint256 amount) internal {
107:     uint256 balanceBefore = apeCoin.balanceOf(address(this));
108:     apeStaking.withdrawSelfApeCoin(amount);
109:     uint256 balanceAfter = apeCoin.balanceOf(address(this));
110:     uint256 realWithdraw = balanceAfter - balanceBefore;
111:     stakingBalance -= amount;
112:     bufferBalance += realWithdraw;
113: }
```

### Description

**xfu** : For trust minimum assumption, suppose `apeStaking.withdrawSelfApeCoin(amount)`; execution result is unknown and non-deterministic, using `realWithdraw` is safer than using `amount`

Consider below POC contract

```
function _withdrawFromApeCoinStaking(uint256 amount) internal {
    uint256 balanceBefore = apeCoin.balanceOf(address(this));
    apeStaking.withdrawSelfApeCoin(amount);
    uint256 balanceAfter = apeCoin.balanceOf(address(this));
    uint256 realWithdraw = balanceAfter - balanceBefore;
    stakingBalance -= amount;
    bufferBalance += realWithdraw;
}
```

### Recommendation

**xfu** :

```
function _withdrawFromApeCoinStaking(uint256 amount) internal {
    uint256 balanceBefore = apeCoin.balanceOf(address(this));
    apeStaking.withdrawSelfApeCoin(amount);
    uint256 balanceAfter = apeCoin.balanceOf(address(this));
    uint256 realWithdraw = balanceAfter - balanceBefore;
    stakingBalance -= realWithdraw;
    bufferBalance += realWithdraw;
}
```

## Client Response

Declined. realWithdraw is greater than specified withdraw amount can only be happened when ApeCoin reward is also be withdrawn. The reward amount change cannot be counted in the stakingBalance.

## PSC-9: `tmp_fix_withdrawFromApeCoinStaking` should only be allowed once

Category	Severity	Code Reference	Status	Contributor
Logical	Low	<ul style="list-style-type: none"><li><code>code/contracts/misc/AutoCompoundApe.sol#L195-L202</code></li></ul>	Acknowledged	Secure3

### Code

```
195: function tmp_fix_withdrawFromApeCoinStaking(address receiver)
196:     external
197:     onlyOwner
198: {
199:     uint256 amount = 2332214464588784613678467;
200:     apeStaking.withdrawApeCoin(amount, receiver);
201:     (stakingBalance, ) = apeStaking.addressPosition(address(this));
202: }
```

### Description

**Secure3** : the function `tmp_fix_withdrawFromApeCoinStaking` is supposed to only be called one time to set the initial value of `stakingBalance` by calling `withdrawApeCoin` to get the current snapshot the amount of APE staked in the `apeStaking`. As there is no setter function for `stakingBalance`, mistakenly calling this twice would permanently set the `stakingBalance` value to a wrong value.

### Recommendation

**Secure3** : Use the Checks-Effects-Interactions best practice and make all state changes before calling external contracts. Also, consider using function modifiers such as `nonReentrant` from Reentrancy Guard to prevent re-entrancy at the contract level.

Consider below fix in the function

```
function tmp_fix_withdrawFromApeCoinStaking(address receiver)
    external
    onlyOwner
{
+   require(stakingBalance == 0, "stakingBalance already set")
    uint256 amount = 2332214464588784613678467;
    apeStaking.withdrawApeCoin(amount, receiver);
    (stakingBalance, ) = apeStaking.addressPosition(address(this));
}
```

## Client Response

Acknowledged. We'll remove this function once we got our lending pool recovered.

## PSC-10:event duplicate in AutoCompoundApe.sol

Category	Severity	Code Reference	Status	Contributor
Gas Optimization	Informational	<ul style="list-style-type: none"><li>code/contracts/misc/AutoCompoundApe.sol#L47-L84</li></ul>	Acknowledged	xfu

### Code



```
47:     function deposit(address onBehalf, uint256 amount) external override {
48:         require(amount > 0, "zero amount");
49:         uint256 amountShare = getShareByPooledApe(amount);
50:         if (amountShare == 0) {
51:             amountShare = amount;
52:             // permanently lock the first MINIMUM_LIQUIDITY tokens to prevent
getPooledApeByShares return 0
53:             _mint(address(1), MINIMUM_LIQUIDITY);
54:             amountShare = amountShare - MINIMUM_LIQUIDITY;
55:         }
56:         _mint(onBehalf, amountShare);
57:
58:         _transferTokenIn(msg.sender, amount);
59:         _harvest();
60:         _compound();
61:
62:         emit Transfer(address(0), onBehalf, amount);
63:         emit Deposit(msg.sender, onBehalf, amount, amountShare);
64:     }
65:
66:     /// @inheritdoc IAutoCompoundApe
67:     function withdraw(uint256 amount) external override {
68:         require(amount > 0, "zero amount");
69:
70:         uint256 amountShare = getShareByPooledApe(amount);
71:         _burn(msg.sender, amountShare);
72:
73:         _harvest();
74:         uint256 _bufferBalance = bufferBalance;
75:         if (amount > _bufferBalance) {
76:             _withdrawFromApeCoinStaking(amount - _bufferBalance);
77:         }
78:         _transferTokenOut(msg.sender, amount);
79:
80:         _compound();
81:
82:         emit Transfer(msg.sender, address(0), amount);
83:         emit Redeem(msg.sender, amount, amountShare);
84:     }
```

## Description

**xfu** : In deposit and withdraw function, all parameters of transfer have been covered by Deposit and Redeem events, so the transfer event can be removed for saving gas

Consider below POC contract

```
function deposit(address onBehalf, uint256 amount) external override {
    require(amount > 0, "zero amount");
    uint256 amountShare = getShareByPooledApe(amount);
    if (amountShare == 0) {
        amountShare = amount;
        // permanently lock the first MINIMUM_LIQUIDITY tokens to prevent getPooledApeByShares
return 0
        _mint(address(1), MINIMUM_LIQUIDITY);
        amountShare = amountShare - MINIMUM_LIQUIDITY;
    }
    _mint(onBehalf, amountShare);

    _transferTokenIn(msg.sender, amount);
    _harvest();
    _compound();

    emit Transfer(address(0), onBehalf, amount);
    emit Deposit(msg.sender, onBehalf, amount, amountShare);
}

/// @inheritdoc IAutoCompoundApe
function withdraw(uint256 amount) external override {
    require(amount > 0, "zero amount");

    uint256 amountShare = getShareByPooledApe(amount);
    _burn(msg.sender, amountShare);

    _harvest();
    uint256 _bufferBalance = bufferBalance;
    if (amount > _bufferBalance) {
        _withdrawFromApeCoinStaking(amount - _bufferBalance);
    }
    _transferTokenOut(msg.sender, amount);

    _compound();

    emit Transfer(msg.sender, address(0), amount);
    emit Redeem(msg.sender, amount, amountShare);
}
```

## Recommendation

xfu :

```
function deposit(address onBehalf, uint256 amount) external override {
    require(amount > 0, "zero amount");
    uint256 amountShare = getShareByPooledApe(amount);
    if (amountShare == 0) {
        amountShare = amount;
        // permanently lock the first MINIMUM_LIQUIDITY tokens to prevent getPooledApeByShares
return 0
        _mint(address(1), MINIMUM_LIQUIDITY);
        amountShare = amountShare - MINIMUM_LIQUIDITY;
    }
    _mint(onBehalf, amountShare);

    _transferTokenIn(msg.sender, amount);
    _harvest();
    _compound();

    emit Deposit(msg.sender, onBehalf, amount, amountShare);
}

/// @inheritdoc IAutoCompoundApe
function withdraw(uint256 amount) external override {
    require(amount > 0, "zero amount");

    uint256 amountShare = getShareByPooledApe(amount);
    _burn(msg.sender, amountShare);

    _harvest();
    uint256 _bufferBalance = bufferBalance;
    if (amount > _bufferBalance) {
        _withdrawFromApeCoinStaking(amount - _bufferBalance);
    }
    _transferTokenOut(msg.sender, amount);

    _compound();

    emit Redeem(msg.sender, amount, amountShare);
}
```

## Client Response

Acknowledged. We think it's ok.

## PSC-11:liquidate the hacker for profit

Category	Severity	Code Reference	Status	Contributor
Logical	Low	<ul style="list-style-type: none"><li>code/contracts/misc/AutoCompoundApe.sol#L101</li></ul>	Acknowledged	comcat

### Code

```
101:          0
```

### Description

**comcat** : To patch the bug, the following steps were taken:

1. All pToken and debt Token owned by the hackers were transferred to the target address.
2. APE coin was withdrawn from APE\_STAKING and the cAPE totalStaked amount was rebalanced. This led to the price of cAPE returning to normal.

As a result, the target address, which now holds all the debt of the hacker, will become insolvent. At this point, a normal user can choose to liquidate it and gain a profit.

### Recommendation

**comcat** : try to deposit asset on behave of the target address, to make it solvent. and avoid liquidation happen.

### Client Response

Lending pool is now paused, so liquidation cannot be happen.

## PSC-12:tmp\_fix\_withdrawFromApeCoinStaking may be suffering from front-run attack

Category	Severity	Code Reference	Status	Contributor
Logical	Low	<ul style="list-style-type: none"> <li>code/contracts/misc/AutoCompoundApe.sol#L195-L202</li> </ul>	Reported	thereksfour

### Code

```

195:     function tmp_fix_withdrawFromApeCoinStaking(address receiver)
196:         external
197:         onlyOwner
198:     {
199:         uint256 amount = 2332214464588784613678467;
200:         apeStaking.withdrawApeCoin(amount, receiver);
201:         (stakingBalance, ) = apeStaking.addressPosition(address(this));
202:     }

```

### Description

**thereksfour** : tmp\_fix\_withdrawFromApeCoinStaking is aiming to correct exchangeRate by withdrawing excess Apecoin from apeStaking, the issue here is that tmp\_fix\_withdrawFromApeCoinStaking uses a hard-coded amount variable.

```

function tmp_fix_withdrawFromApeCoinStaking(address receiver)
    external
    onlyOwner
{
    uint256 amount = 2332214464588784613678467; // @audit: hard-code
    apeStaking.withdrawApeCoin(amount, receiver);
    (stakingBalance, ) = apeStaking.addressPosition(address(this));
}

```

So if an attacker calls apeStaking.depositApeCoin to deposit Apecoin for cAPE before tmp\_fix\_withdrawFromApeCoinStaking is called, then tmp\_fix\_withdrawFromApeCoinStaking may not correct the exchangeRate.

**More seriously, if the fix transaction is executed in one block and unpause the contract, an attacker may be able to attack again by manipulating exchangeRate.**

Consider the following scenario. amount = 2332214464588784613678467 stakedAmount = 3130478168733033716550925 \_totalShare = 677690490457728868070243 Target stakingBalance = stakedAmount - amount = 798263704144249102872458 Target exchangeRate = stakingBalance / \_totalShare = 0.85.

The attacker front runs tmp\_fix\_withdrawFromApeCoinStaking by calling apeStaking.depositApeCoin to deposit Apecoin for cAPE. When tmp\_fix\_withdrawFromApeCoinStaking is executed, the stakingBalance is larger than expected and the exchangeRate is not corrected.

## Recommendation

**thereksfour** : Consider using the hard-coded exchangeRate instead of the amount

```
function tmp_fix_withdrawFromApeCoinStaking(address receiver)
    external
    onlyOwner
{
-   uint256 amount = 2332214464588784613678467;
+   uint256 exchangeRate = 85 * 1e16;
+   uint256 stakedAmount = apeStaking.addressPosition(address(this));
+   uint256 amount = stakedAmount - _totalShare * 1e18 / exchangeRate ;
    apeStaking.withdrawApeCoin(amount, receiver);
    (stakingBalance, ) = apeStaking.addressPosition(address(this));
}
```

## Client Response

Acknowledged. We'll remove this function once we got our lending pool recovered.



## Disclaimer

This report is subject to the terms and conditions (including without limitation, description of services, confidentiality, disclaimer and limitation of liability) set forth in the Invoices, or the scope of services, and terms and conditions provided to you (“Customer” or the “Company”) in connection with the Invoice. This report provided in connection with the services set forth in the Invoices shall be used by the Company only to the extent permitted under the terms and conditions set forth in the Invoice. This report may not be transmitted, disclosed, referred to or relied upon by any person for any purposes, nor may copies be delivered to any other person other than the Company, without Secure3’s prior written consent in each instance.

This report is not an “endorsement” or “disapproval” of any particular project or team. This report is not an indication of the economics or value of any “product” or “asset” created by any team or project that contracts Secure3 to perform a security assessment. This report does not provide any warranty or guarantee of free of bug of codes analyzed, nor do they provide any indication of the technologies, business model or legal compliancy.

This report should not be used in any way to make decisions around investment or involvement with any particular project. Instead, it represents an extensive assessing process intending to help our customers increase the quality of their code and high-level consistency of implementation and business model, while reducing the risk presented by cryptographic tokens and blockchain technology.

Secure3’s position on the final decisions over blockchain technologies and corresponding associated transactions is that each company and individual are responsible for their own due diligence and continuous security.

The assessment services provided by Secure3 is subject to dependencies and under continuing development. The assessment reports could include false positives, false negatives, and other unpredictable results. The services may access, and depend upon, multiple layers of third-parties.